Comma quibbling is a task originally set by Eric Lippert in his blog.



Task

Write a function to generate a string output which is the concatenation of input words from a list/sequence where:

 An input of no words produces the output string of just the two brace characters "{}".
 An input of just one word, e.g. ["ABC"], produces the output string of the word inside the two braces, e.g. "{ABC}".
 An input of two words, e.g. ["ABC", "DEF"], produces the output string of the two words inside the two braces with the words separated by the string " and ", e.g. "{ABC and DEF}".
 An input of three or more words, e.g. ["ABC", "DEF", "G", "H"], produces the output string of all but the last word separated by ", " with the last word separated by " and " and all within braces; e.g. "{ABC, DEF, G and H}".


Test your function with the following series of inputs showing your output here on this page:

 [] # (No input words).
 ["ABC"]
 ["ABC", "DEF"]
 ["ABC", "DEF", "G", "H"]


Note: Assume words are non-empty strings of uppercase characters for this task.




quibble = proc (words: array[string]) returns (string)
    out: string := "{"
    last: int := array[string]$high(words)
 
    for i: int in array[string]$indexes(words) do
        out := out || words[i]
        if i < last-1 then
            out := out || ", "
        elseif i = last-1 then
            out := out || " and "
        end
    end
    return(out || "}")
end quibble
 
start_up = proc ()
    as = array[string]
    aas = array[as]
    po: stream := stream$primary_output()
 
    testcases: aas := aas$
        [as$[],
         as$["ABC"],
         as$["ABC","DEF"],
         as$["ABC","DEF","G","H"]]
 
    for testcase: as in aas$elements(testcases) do
        stream$putl(po, quibble(testcase))
    end 
end start_up